Applications of MPEC models

Steven Dirkse

GAMS Development Corporation

INFORMS-2003 Atlanta
October 19-22, 2003

Background & Motivation

- Academic interest in a new model type
  - Interesting challenge to add this to GAMS
  - Natural extension of complementarity
- Practical/commercial interest
  - Growing literature on applications & solution methods
  - Customer demand – GAMS users asked for this
- Experiments with solution methods

Introduction

- Background and motivation
- Applications
  - Inverse/identification problems in engineering
  - Network design – traffic model
  - Game theory

MPEC Definition

$$\min \quad f(x,y)$$

$$\text{s.t.} \quad g(x,y) \geq 0, \quad L_x \leq x \leq U_x$$

$$y \quad \text{solves} \quad \text{MCP} \;\; (h(x,\cdot\,), B)$$

- The NLP constraints g(x,y) are well understood
- y solves MCP(h,B) - equilibrium constraint

MCP Definition

y solves MCP $(h(x,\cdot\,), B)$ where $B = [L, U]$

$$\Leftrightarrow$$

$$h_i(x, y) \quad \perp \quad L_i \le y_i \le U_i \qquad \forall i$$

$$\Leftrightarrow$$

$h_i(x, y) = 0$ or

$h_i(x, y) \ge 0$ and $L_i = y_i$ or

$h_i(x, y) \le 0$ and $y_i = U_i$

Identification Problems

Real Ident Problem

- Elastoplastic beam, elastoplastic foundation
- Given material properties, MCP provides:
  – Position data
  – Compression and pressure data
- Given measurements of position, solve MPEC to:
  – Minimize || computed pos – measured pos ||
  – s.t. bounds on material properties
    equil. cons.

GAMS Source

```
equil(dof)..sum(members,C(members,dof)*QF(members)) =e= F(dof);
constit(members)..
   QF(members)/S(members) =e= sum(dof,C(members,dof)*u(dof))
   - lambda("plus",members) + lambda("minus",members);
yield(plastic,members)..
   - QF(members)$plus(plastic) + QF(members)$minus(plastic)
   + q0(plastic,"beam")$be(members)
   + q0(plastic,"spring")$sp(members)
   + (h(plastic,"beam")*lambda(plastic,members))$be(members)
   + (h(plastic,"spring")*lambda(plastic,members))$sp(members)
   =g= 0;
q0.fx(...) = ...;  h.fx(...) = 0;
model base /equil.u,constit.QF,yield.lambda /;
PARAMETER pertu(dof) / …. /;
variables mu;
positive variables uerr(dof1);
equations lower(dof1),upper(dof1),uerrdef;
lower(dof1).. u(dof1) =g= pertu(dof1) - uerr(dof1);
upper(dof1).. u(dof1) =l= pertu(dof1) + uerr(dof1);
uerrdef.. mu =e= sum(dof1,uerr(dof1));
model identify /equil.u,constit.QF,yield.lambda,lower,upper,uerrdef/;
```

Real Ident Problem II

- References
  - Maier et al. (Eng. Struct. 4, 1982, 86-98)
  - Grierson, Chapter 14, Waterloo Workshop
  - F. Tin-Loi
- The real model contains lots of detail – this can hide what’s going on
- A small example illustrates the main points

Simple Ident Problem

- Given MCP(f,B) parameterized by p.
- Observe a component of the solution value
- Find p s.t. equilibrium solution matches observation
- Bounds on p mean MCP may not solve – MPEC computes a “nearby” solution

Simple Ident Problem II

```
scalar xbar / 1 /;
variable z, p;
positive variable x, y;
equation f1, f2, eDef;
f1..sqr(x)+sqr(y) =e= sqr(p);
f2..y =g= log(x);
eDef..   z =e= sqr(x-xbar);
model m / eDef, f1, f2.y /;
p.lo = .7; p.up = 1.3; x.l = .1;
solve m using mpec min z;
```

Simple Ident Problem III



Model results



Network design (Rutherford,Dean)

- Assume a city of neighborhoods (nodes) located on a square grid
- Traffic links join nodes to NEWS neighbors
- Each node has
  - Employment with wage rate w(j), demand curve relating w(j) to demand
  - Fixed housing stock, housing price ph(j)
  - Flow balancing equations

Network design II

- Each arc has
  - traffic flow F(i,j) and resulting delay(i,j)
  - X(i,j,k) – portion of F(i,j) destined for node k
- Each path (i,j) has
  - N(i,j) commuting from i to workplace j
  - Travel time T(i,j)
  - Utility function V(T(i,j),ph(i),w(j))
- We have an overall utility U

Network Equilibrium Conditions

$$\text{delay}_{ij} + T_{jk} \geq T_{ik} \;\; \forall i, j, k \quad \perp \quad X_{ijk} \geq 0$$

$$U \geq V(T_{ik}, p_i^H, w_j) \quad \perp \quad N_{ij} \geq 0$$

- Flow balance constraints
- Demand equations

Benchmark – no commuting



Counterfactuals

- Case I: concentrate jobs in the NE corner
  - Increases housing prices in NE
  - Increases Vehicle Miles Traveled
- Case II: same job concentration, but minimize additional housing needed to cut Case I VMT by 50%.
  - Less drastic increase in housing prices
  - Increases population in the NE

Case I – jobs in NE



Case II – % New Housing



Case I – % Pop. Density Change



Case II – % Pop. Density Change

Population Density w/ new housing



Case I – Housing Price

Case II – Housing Price

Housing Prices w/ new housing



The Stranger

- “Say you're in a public library, and a beautiful stranger strikes up a conversation with you. She says: ‘Let's show pennies to each other, either heads or tails. If we both show heads, I pay you $3. If we both show tails, I pay you $1. If they don't match, you pay me $2.’ At this point, she is shushed. You think, ‘With both heads 1/4 of the time, I get $3. And with both tails 1/4 of the time, I get $1. So 1/2 of the time, I get $4. And with no matches 1/2 of the time, she gets $4. So it's a fair game.’"

The Stranger II

- "As the game is quiet, you can play in the library. But should you? Should she? -- Edward Spellman, Cheshire, Connecticut"
- "Ask Marilyn", Parade, March 31 2002,
- SIAM News, June 2003.

Modeling the game

```
SET s 'pure strategies' / heads, tails /;
ALIAS (s,you,her);
TABLE herCost (you,her)
         heads   tails
heads      3      -2
tails     -2        1   ;

PARAMETER yourCost(you,her);
yourCost(you,her) = -herCost(you,her);
```

Modeling the game II

```
POSITIVE VARIABLES
x(you)    'your mixed strategy',
y(her)    'her mixed strategy';

FREE VARIABLES
u 'your multiplier-expected loss',
v 'her multiplier-expected loss';
```

Modeling the game III

```
fYou(you)..sum{her, yourCost(you,her)*y(her)}
           - u =g= 0;
sumYou..   sum{you, x(you)} =e= 1;

fHer(her)..sum{you, x(you)*herCost(you,her)}
           - v =g= 0;
sumHer..   sum{her, y(her)} =e= 1;

model nashEquil / fYou.x, sumYou.u,
                  fHer.y, sumHer.v /;
```

Finding a Nash Equilibrium

```
solve nashEquil using mcp;
```

- Your x('heads','tails') = (3/8, 5/8)
- Her y('heads','tails') = (3/8, 5/8)
- Expected winner: her, $.125/game.

Finding Her Winning Range

- For what values of y(heads) is she safe?

```
lDef..loss =e= sum{(you,her),
      x(you)*yourCost(you,her)*y(her)};
zDef..  z =e= y('heads');
model m /lDef, fYou.x, sumYou.u,
         sumHer, zDef/;
loss.lo = 0;
```

Finding Her Winning Range

```
x.up(you) = 1.0; y.up(her) = 1.0;
u.lo = -3;  u.up = 3;
v.lo = -3;  v.up = 3;
option nlp = baron;
solve m using mpec min z;
solve m using mpec max z;
```

- Y(heads) in [1/3,.4] is safe (3/8 is optimal)

Playing for Charity

- What if she says "I'll pay some extra for the tails-tails case, but the extra goes to charity". How much can she pay extra and still break even?

```
parameter t(you,her) /tails.tails 1/;
variable r 'charity tax rate from T-T';
fHer2(her)..   sum{you,  x(you)
  *[herCost(you,her)+t(you,her)*r]} - v =g= 0;
model m2 / fYou.x, sumYou.u,
           fHer2.y, sumHer /;
v.up = 0;
solve m2 max r using mpec;
```

- Solution: r = 1/3 extra, but only your strategy changes!

Conclusions

- MPEC models provide a flexible and powerful way to model many things
- Add additional value to many client models
- Solving these models is no longer a job for a "domain expert"
- `http://www.gams.com/presentations`
- `http://www.gamsworld.org/mpec`